2ExtensionModules

In this worksheet several different examples for computing $\mathrm{ExtMod}(\eta_L^M, \eta_N^L)$ from two 1-cocycles $\eta_L^M \in \mathrm{Ext}^1(M, L)$ and $\eta_N^L \in \mathrm{Ext}^1(L, N)$ with `homalg` will be given. They cover different cases that can occur, in particular rigid and non-rigid pairs of 1-cocycles (cf. [BB]).

```
> restart;
> with(homalg): with(Involutive):
```

Specify the ring package `Involutive` as the default ring package, it allows the computation over polynomial rings.

```
> `homalg/default`:=`Involutive/homalg`;
```

$$homalg/default := Involutive/homalg$$

Let $D = \mathbb{Q}[x, y, z]$, this is specified in the variable `var`.

```
> var:=[x,y,z];
```

$$var := [x,\ y,\ z]$$

1. Example 1

Consider the modules $M = \mathrm{coker}(\mathtt{M})$, $L = \mathrm{coker}(\mathtt{L})$ and $N = \mathrm{coker}(\mathtt{N})$ as follows:

```
> M:=[x,y];
```

$$M := [x,\ y]$$

```
> L := [[x, y], [0, x*y-z]];
```

$$L := [[x,\ y],\ [0,\ x\,y - z]]$$

```
> N:=[y,x,z^2];
```

$$N := [y,\ x,\ z^2]$$

`homalg` can compute $\mathrm{Ext}^1(M, L)$ and $\mathrm{Ext}^1(L, N)$:

```
> Ext(1,M,L,var,"var_to_assign_proc_to_express_generators_abstractly"='
abstractlyML');
```

$$[[1 = \begin{bmatrix} 1 & 0 \\ 0 & -1 \end{bmatrix}],\ [y,\ x],\ \text{“Presentation”},\ 1 + \frac{s}{1-s},\ [1,\ 0,\ 0]]$$

```
> Ext(1,L,N,var,"var_to_assign_proc_to_express_generators_abstractly"='
abstractlyLN');
```

$$[[[1,\ 0] = \begin{bmatrix} 0 \\ 1 \end{bmatrix},\ [0,\ 1] = \begin{bmatrix} 1 \\ 0 \end{bmatrix}],\ [[z,\ 0],\ [0,\ y],\ [y,\ 0],\ [0,\ x],\ [x,\ 0],\ [0,\ z^2]],\ \text{“Presentation”},\ s+2,\ [0,\ 0,\ 0]]$$

Choose 1-cocycles η_L^M and η_N^L that are not trivial. The procedures `abstractlyML` and `abstractlyLN` assigned in the computation of the Ext-groups calculate an abstract representation of a given 1-cocycle, i.e. the corresponding linear combination of the generators of $\mathrm{Ext}^1(M, L)$ and $\mathrm{Ext}^1(L, N)$ computed above. We use them to check that the 1-cocycles are indeed not trivial.

```
> etaML:=matrix([[z, 0], [0, -z]]);
```

$$etaML := \begin{bmatrix} z & 0 \\ 0 & -z \end{bmatrix}$$

```
> abstractlyML(etaML);
```

$$z$$

```
> etaLN:=matrix([[z], [0]]);
```

$$etaLN := \begin{bmatrix} z \\ 0 \end{bmatrix}$$

```
> abstractlyLN(etaLN);
```

$$[0,\ z]$$

The YONEDA product of these cocycles vanishes:

```
>  YonedaProductOfCocycles(M,etaML,1,L,etaLN,1,N,var,"return_abstract_ge
nerator");
```

$$0 = \left[\ z^2\ \right]$$

Even though the matrix (z^2) representing the product is not zero, the abstract representation shows it to be trivial in $\mathrm{Ext}^2(M,N)$. Therefore we know that $\mathrm{ExtMod}(\eta_L^M, \eta_N^L) \neq \emptyset$ and compute a particular solution with the procedure `A2ExtensionModule`.

```
>  eta := A2ExtensionModule(M,etaML,L,etaLN,N,var);
```

$$\eta := \begin{bmatrix} 0 \\ 0 \end{bmatrix}$$

So with this η we found a 2-extension module $E = coker(\mathtt{E})$ with relation matrix

$$\mathtt{E} = \begin{pmatrix} \mathtt{M} & \eta_L^M & \begin{pmatrix} 0 \\ 0 \end{pmatrix} \\ \cdot & \mathtt{L} & \eta_N^L \\ \cdot & \cdot & \mathtt{N} \end{pmatrix}.$$

`Ext1Of2OneCocycles` computes the first extension group of 1-cocycles

$$\mathrm{Ext}^1(\eta_L^M, \eta_N^L) \ = \mathrm{Ext}^1(M,N)/(\mathrm{Hom}(M,L) \circ \eta_N^L + \eta_L^M \circ \mathrm{Hom}(L,N)),$$

in this case it is not trivial:

```
>  Ext1Of2OneCocycles(M,etaML,L,etaLN,N,var);
```

$$[[[1,\ 0] = \begin{bmatrix} 0 \\ 1 \end{bmatrix},\ [0,\ 1] = \begin{bmatrix} 1 \\ 0 \end{bmatrix}],\ [[0,\ z],\ [0,\ y],\ [y,\ 0],\ [0,\ x],\ [x,\ 0],\ [z^2,\ 0]],\ \text{“Presentation”},\ s+2,\ [0,\ 0,\ 0]]$$

The group is not $\mathrm{Ext}^1(M,N)$ itself, but a proper factor ($z \left(\begin{smallmatrix} 1 \\ 0 \end{smallmatrix}\right)$ is trivial here, it is not trivial in $\mathrm{Ext}^1(M,N)$):

```
>  Ext(1,M,N,var);
```

$$[[[1,\ 0] = \begin{bmatrix} 0 \\ 1 \end{bmatrix},\ [0,\ 1] = \begin{bmatrix} 1 \\ 0 \end{bmatrix}],\ [[0,\ y],\ [y,\ 0],\ [0,\ x],\ [x,\ 0],\ [0,\ z^2],\ [z^2,\ 0]],\ \text{“Presentation”},\ 2+2\,s,\ [0,\ 0,\ 0]]$$

So all possible 2-extension modules correspond to $\eta + \eta_N^M$, where η as above and $\eta_N^M \in \mathrm{Ext}^1(\eta_L^M, \eta_N^L)$. In particular it is immediate that (η_L^M, η_N^L) is non-rigid.

2. EXAMPLE 2

This is the example of Subsection 8.2 in [BB].
Again we choose modules and non-trivial 1-cocycles:

```
>  M := [x,y,z];
```

$$M := [x,\ y,\ z]$$

```
>  L := [x^5,z];
```

$$L := [x^5,\ z]$$

```
>  N := [x^5,z];
```

$$N := [x^5,\ z]$$

```
>  Ext(1,M,L,var,"var_to_assign_proc_to_express_generators_abstractly"='
abstractlyML');
```

$$\left[\left[1 = \begin{bmatrix} 0 \\ x^4 \\ 0 \end{bmatrix}\right],\ [z,\ y,\ x],\ \text{“Presentation”}\ ,\ 1,\ [0,\ 0,\ 0]\right]$$

```
> Ext(1,L,N,var,"var_to_assign_proc_to_express_generators_abstractly"='
abstractlyLN');
```

$$[[[1,\,0] = \left[\begin{array}{c}0\\1\end{array}\right],\,[0,\,1] = \left[\begin{array}{c}1\\0\end{array}\right]],[[0,\,z],\,[z,\,0],\,[0,\,x\,z],\,[x\,z,\,0],\,[0,\,x^2\,z],\,[x^2\,z,\,0],\,[0,\,x^3\,z],\,[x^3\,z,\,0],\,[0,\,x^4\,z],\,[x^4\,z,\,0],\,[0,\,x^5],\,[x^5,\,0]],\,\text{“Presentation”},\,2+4\,s+6\,s^2+8\,s^3+10\,s^4+\frac{10\,s^5}{1-s},\,[10,\,0,\,0]]$$

```
> etaML := matrix([[0], [x^4], [0]]);
```

$$etaML := \left[\begin{array}{c}0\\x^4\\0\end{array}\right]$$

```
> abstractlyML(etaML);
```

$$1$$

```
> etaLN := matrix([[0], [x]]);
```

$$etaLN := \left[\begin{array}{c}0\\x\end{array}\right]$$

```
> abstractlyLN(etaLN);
```

$$[x,\,0]$$

The YONEDA product vanishes,

```
> YonedaProductOfCocycles(M,etaML,1,L,etaLN,1,N,var,"return_abstract_ge
nerator");
```

$$[0,\,0] = \left[\begin{array}{c}0\\x\\0\end{array}\right]$$

so $\mathrm{ExtMod}(\eta_L^M, \eta_N^L) \neq \emptyset$. A particular solution is

```
> eta := A2ExtensionModule(M,etaML,L,etaLN,N,var);
```

$$\eta := \left[\begin{array}{c}0\\1\\0\end{array}\right]$$

```
> Ext(1,M,N,var,"var_to_assign_proc_to_express_generators_abstractly"='
abstractlyMN');
```

$$\left[\left[\left[1 = \left[\begin{array}{c}0\\x^4\\0\end{array}\right]\right]\right],\,[z,\,y,\,x],\,\text{“Presentation”}\,,\,1,\,[0,\,0,\,0]\right]$$

```
> abstractlyMN(eta);
```

```
Error, (in homalg/RightDivide) The second argument is not a right
factor of the first (modulo the third), i.e. the second (+third)
argument is not a generating set!
```

This means η can not be written as a linear combination of the generators of $\mathrm{Ext}^1(M,N)$ and is therefore not an element of this group. This also means that the zero matrix is not a valid solution for this choice of representatives of η_L^M and η_N^L.

The group modelling the solution space is trivial:

```
> Ext1Of2OneCocycles(M,etaML,L,etaLN,N,var);
```

$$\left[\left[\left[1 = \left[\begin{array}{c}0\\x^4\\0\end{array}\right]\right]\right],\,[1],\,\text{“Presentation”},\,0,\,[0,\,0,\,0]\right]$$

so $|\,\mathrm{ExtMod}(\eta_L^M, \eta_N^L)| = 1$ and (η_L^M, η_N^L) is rigid.

3. Example 3

```
> M := [x,y,z];
```

$$M := [x,\, y,\, z]$$

```
> L := [x^5,z];
```

$$L := [x^5,\, z]$$

```
> N := [x^3,z];
```

$$N := [x^3,\, z]$$

```
> Ext(1,M,L,var,"var_to_assign_proc_to_express_generators_abstractly"='
abstractlyML');
```

$$\left[\left[1 = \begin{bmatrix} 0 \\ x^4 \\ 0 \end{bmatrix}\right],\ [z,\, y,\, x],\ \text{“Presentation”}\ ,\ 1,\ [0,\, 0,\, 0]\right]$$

```
> Ext(1,L,N,var,"var_to_assign_proc_to_express_generators_abstractly"='
abstractlyLN');
```

$$[[[1,\, 0] = \begin{bmatrix} 0 \\ 1 \end{bmatrix},\ [0,\, 1] = \begin{bmatrix} 1 \\ 0 \end{bmatrix}],\ [[0,\, z],\ [z,\, 0],\ [0,\, x\, z],\ [x\, z,\, 0],\ [0,\, x^2\, z],\ [x^2\, z,\, 0],\ [0,\, x^3],\ [x^3,\, 0]],$$

$$\text{“Presentation”},\ 2 + 4\, s + 6\, s^2 + \frac{6\, s^3}{1 - s},\ [6,\, 0,\, 0]]$$

```
> etaML := matrix([[0], [x^4], [0]]);
```

$$etaML := \begin{bmatrix} 0 \\ x^4 \\ 0 \end{bmatrix}$$

```
> abstractlyML(etaML);
```

$$1$$

```
> etaLN := matrix([[0], [x]]);
```

$$etaLN := \begin{bmatrix} 0 \\ x \end{bmatrix}$$

```
> abstractlyLN(etaLN);
```

$$[x,\, 0]$$

The Yoneda product vanishes

```
> YonedaProductOfCocycles(M,etaML,1,L,etaLN,1,N,var,"return_abstract_ge
nerator");
```

$$[0,\, 0] = \begin{bmatrix} 0 \\ x \\ 0 \end{bmatrix}$$

and a particular solution is

```
> eta := A2ExtensionModule(M,etaML,L,etaLN,N,var);
```

$$\eta := \begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix}$$

```
> Ext(1,M,N,var,"var_to_assign_proc_to_express_generators_abstractly"='
abstractlyMN');
```

$$\left[\left[1 = \begin{bmatrix} 0 \\ x^2 \\ 0 \end{bmatrix}\right],\ [z,\, y,\, x],\ \text{“Presentation”}\ ,\ 1,\ [0,\, 0,\, 0]\right]$$

```
> abstractlyMN(eta);
```

```
Error, (in homalg/RightDivide) The second argument is not a right
factor of the first (modulo the third), i.e. the second (+third)
argument is not a generating set!
```

As in the previous example, η is not an element of $\mathrm{Ext}^1(M, N)$, so the zero matrix is not a solution. However, this time the group $\mathrm{Ext}^1(\eta_L^M, \eta_N^L)$ modelling the solution space is not trivial, so (η_L^M, η_N^L) is non-rigid. In fact, $\mathrm{Ext}^1(\eta_L^M, \eta_N^L) = \mathrm{Ext}^1(M, N)$:

```
> Ext1Of2OneCocycles(M,etaML,L,etaLN,N,var);
```

$$\left[\left[1 = \left[\begin{array}{c} 0 \\ x^2 \\ 0 \end{array}\right]\right], [z, y, x], \text{“Presentation”}, 1, [0, 0, 0]\right]$$

Author: Mohamed Barakat and Barbara Bremer

Date:	2008-02-27
Last modified:	2008-02-27 16:39

References

[BB] Mohamed Barakat and Barbara Bremer, *Higher Extension Modules and the Yoneda Product*, submitted http://wwwb.math.rwth-aachen.de/homalg). 1, 2

[BCG+03] Y. A. Blinkov, C. F. Cid, V. P. Gerdt, W. Plesken, and D. Robertz, *The* MAPLE *Package* Janet*: I. Polynomial Systems. II. Linear Partial Differential Equations*, Proc. 6th Int. Workshop on Computer Algebra in Scientific Computing, Passau, Germany, 2003, (http://wwwb.math.rwth-aachen.de/Janet), pp. 31–40 and 41–54.

[BR] Mohamed Barakat and Daniel Robertz, homalg *– A meta-package for homological algebra*, accepted for publication in Journal of Algebra and its Applications. (arXiv:math.AC/0701146 and http://wwwb.math.rwth-aachen.de/homalg).

[BR08] ———, homalg *project*, 2003-2008, (http://wwwb.math.rwth-aachen.de/homalg).

Lehrstuhl B für Mathematik, RWTH-Aachen University, 52062 Germany
E-mail address: mohamed.barakat@rwth-aachen.de

Lehrstuhl B für Mathematik, RWTH-Aachen University, 52062 Germany
E-mail address: barbara.bremer@rwth-aachen.de